

A conservative extension of ECLAT with ESTEREL and LUSTRE constructs (WIP)

Loïc Sylvestre¹

¹Université de Toulouse - IRIT (TRACES team)

November 26 2025 - SYNCHRON '25 à Aussois

Context

- ▶ The ECLAT programming language
 - functional programming language (*à la OCAML*) with synchronous semantics
 - compiled to VHDL for programming FPGAs.
- ▶ ECLAT initial design choice: keep it simple
 - avoid causality/dependency issues
 - ECLAT programs are **acyclic by construction**
 - and **communications between parallel “threads” are delayed**
(via built-in shared memory accesses with deterministic concurrency)
 - **but this is quite restrictive...**
- ▶ A long-term wish: embedding LUSTRE in ECLAT, to keep it simple but more expressive.
- ▶ Ongoing work, an exploratory path:
 - extend ECLAT with instantaneous signal broadcasting
(and while you are at it, extend ECLAT with a complete ESTEREL kernel)
 - let programmers define their own fixpoint operator and other derivated constructs, in ECLAT, as polymorphic higher-order functions.

Comparing

ECLAT, ESTEREL and LUSTRE

If **not(A)** then **B** and **C** (*ECLAT is closer to LUSTRE*)

ESTEREL


```
module NOTABC:  
  input A;  
  output B, C;  
  loop  
    present A else emit B; emit C end;  
    pause  
  end  
end
```

ECLAT

```
let notabc (a:bool) : bool*bool =  
  let x = not(a) in  
  (x,x) ;;
```

```
node notabc (a:bool)  
  returns (b,c:bool);  
let  
  b = c;  
  c = not(a);  
tel
```

LUSTRE



a	T	T	F	T	F	F	T	F	...
b	F	F	T	F	T	T	F	T	...
c	F	F	T	F	T	T	F	T	...

Wait $N > 0$ instants and terminate (*ECLAT is closer to ESTEREL*)


ESTEREL

```
module WAIT_N;  
  input N : integer;  
  signal S : integer in  
  trap T in  
    pause; emit S(pre(?N));  
    loop  
      if ?S = 1 then exit T  
      else  
        pause; emit S(pre(?S)-1)  
      end  
    end  
  end  
end  
end
```

ECLAT

```
let rec wait_n (n:int) : unit =  
  if n = 1 then () else wait_n(n-1) ;;
```

NB: a pause per direct or tail-recursive call to wait_n



Wait $N > 0$ instants and terminate (*ECLAT is closer to ESTEREL*)

ESTEREL

```
module WAIT_N;  
  input N : integer;  
  signal S : integer in  
  trap T in  
    pause; emit S(pre(?N));  
    loop  
      if ?S = 1 then exit T  
      else  
        pause; emit S(pre(?S)-1)  
      end  
    end  
  end  
end
```

ECLAT

```
let rec wait_n (n:int) : unit =  
  if n = 1 then () else wait_n(n-1) ;;
```

NB: a pause per direct or tail-recursive call to wait_n

```
node wait_n (rst:bool ; n:int)  
  returns (rdy:bool);  
var s : int;  
let  
  s = if (false fby rst) then (0 fby n)  
      else  
        (0 fby (if s = 1 then s else s-1));  
rdy = (not rst) and (s = 1);  
tel
```

LUSTRE

Entry point of the reactive program (*ECLAT is closer to LUSTRE*)

ESTEREL

```
module MAIN;
  input A; N : integer;
  output B, C, RDY;
  [
    run NOTABC
  ||
    loop
      run WAIT_N;
      emit RDY;
      pause
    end
  ]
end
```

ECLAT

```
let main ((n,a):int*bool) : bool*bool*bool =
  let (b,c) = notabc(a) in
  let ((),rdy) = exec wait_n(n) default () in
  (b,c,rdy) ;;
```

```
node main (a:bool ; n:int)
  returns (b,c,rdy:bool);
var rst : bool;
let
  (b,c) = notabc(a);
  rdy = wait_n(rst,n);
  rst = true fby rdy;
tel
```

LUSTRE

ECLAT syntax and semantics

ECLAT core language

program

pattern

expression

simple expressions

sequential composition

parallel composition

conditional

local function definition

function application

register

step-by-step execution

pause and tail-recursion

constant

primitive

$\pi ::= \text{fun } p \rightarrow e$

$p ::= () \mid x \mid (p, p)$

$e ::=$

$x \mid c \mid \phi(e) \mid (e, e)$

$\mid \text{let } p = e \text{ in } e \mid e; e$

$\mid (e \parallel e)$

$\mid \text{if } e \text{ then } e \text{ else } e$

$\mid \text{fun } p \rightarrow e$

$\mid f e$

$\mid \text{reg } (\text{fun } p \rightarrow e) \text{ init } e$

$\mid \text{exec } e \text{ default } e_0$

$\mid \text{pause} \mid \text{let rec } f p = e \text{ in } e$

$c ::= () \mid \text{true} \mid \text{false} \mid n$

$\phi ::= + \mid - \mid < \mid \dots$

Synchronous semantics

program	$\pi ::= \mathbf{fun} \ p \rightarrow e$
pattern	$p \in \mathcal{P}$
expression	$e \in \mathcal{E}$
value	$v \in \mathcal{V} \subseteq \mathcal{E}$
stream	$\omega ::= v \cdot \omega$
memory state	$\mu \in \Sigma$

$$\frac{\pi(v)/\mu \longrightarrow^* v'/\mu' \quad \mu' \vdash \pi(\omega) \Downarrow \omega'}{\mu \vdash \underbrace{(\mathbf{fun} \ p \rightarrow e)}_{\pi}(v \cdot \omega) \Downarrow (v' \cdot \omega')}$$

- ▶ $\mu \vdash \pi(\omega) \Downarrow \omega'$: given memory μ , program π applied to stream ω computes ω'
 - length-preserving function on streams (one input per instant)
 - defined as an instantaneous step function applied at each instant: gets the current input v and returns the current output v' by a finite sequence of reductions (microsteps):

$$\pi(v)/\mu \longrightarrow e_1/\mu_1 \longrightarrow \dots v'/\mu'$$

Reduction semantics, by substitution, with call-by-value evaluation contexts (extracts)

evaluation context $F ::= \mathbf{let} \ p = \square \ \mathbf{in} \ e \mid \square; e \mid \mathbf{exec}_\ell \ e \ \mathbf{default} \ \square \mid \dots$

$$\text{CONTEXT} \quad \frac{e/\mu \longrightarrow e'/\mu'}{F[e]/\mu \longrightarrow F[e']/\mu'}$$

$$\text{LET} \quad \mathbf{let} \ p = v \ \mathbf{in} \ e/\mu \longrightarrow e[p \mapsto v]/\mu$$

$$\text{SEQ} \quad () ; e/\mu \longrightarrow e/\mu$$

$$\text{PAUSE} \quad F[\mathbf{pause}]/\mu \longrightarrow \mathbf{pause}; F[()]/\mu$$

$$\text{PAR} \quad (\mathbf{pause} \ e \parallel \mathbf{pause} \ e')/\mu \longrightarrow \mathbf{pause} \ (e \parallel e')/\mu$$

NB: each time a pause occurs, it is propagated and finally removed by an outer **exec** block at the end of the instant (if the program is well-typed). The continuation of the computation is saved as a thunk (**fun** $() \rightarrow e$) in the memory μ .

Static typing

- ▶ abstracting the execution time (D):
 - always instantaneous ($\mathbf{0}$),
 - 0 or more instants ($\mathbf{1}$).
- ▶ distinguish basic types $\dot{\tau}$ (`unit`, `bool`, `int`, pair of basic types) *versus* types τ (including functional types)
- ▶ functional type of the form $\tau \xrightarrow{D} \dot{\tau}'$
 - limited form of higher-orderness (the return type is a basic type)
 - inferred duration D is used to ensure reactivity, for example:

$$\frac{\text{PROG} \quad \emptyset[p : \dot{\tau}] \vdash e : \dot{\tau}' | \mathbf{0}}{\vdash (\text{fun } p \rightarrow e) : \dot{\tau} \xrightarrow{\mathbf{0}} \dot{\tau}'} \quad \text{PAUSE} \quad \Gamma \vdash \text{pause} : \text{unit} | \mathbf{1}$$

$$\frac{\text{EXEC} \quad \Gamma \vdash e : \dot{\tau} | D \quad \Gamma \vdash e_0 : \dot{\tau} | \mathbf{0}}{\Gamma \vdash \text{exec } e \text{ default } e_0 : \dot{\tau} \times \text{bool} | \mathbf{0}}$$

A conservative extension of ECLAT with ESTEREL constructs

Add the missing constructs

ECLAT	expression	$e ::= \dots$	
	nothing	()	(constant of type unit)
	pause	pause	
	sequence	$e_1; e_2$	
	parallel pair	$(e_1 \parallel e_2)$	syntactic sugar: $[e_1 \parallel e_2]$ stands for $(\text{let } ((), ()) = (e_1 \parallel e_2) \text{ in } ())$
conditional	if e_1 then e_2 else e_3		

conservative extension to support the whole ESTEREL kernel	emission	emit $x(e)$
	reception	? x
	local signal	signal x in e
	loop	loop e end
	trap	trap x in e
	exit	exit x
	suspend	suspend e when x

Valued signals

expression	$e ::= \dots$		signal x in e		emit $x(e)$		$?S$
evaluation context	$F ::= \dots$		emit $s(\square)$				
type	$\tau ::= \dots$		signal $\langle \dot{\tau} \rangle$				
basic type	$\dot{\tau} ::= \dots$						

$$\frac{\Gamma[x : \text{signal}\langle \dot{\tau}_x \rangle] \vdash e : \dot{\tau} | D}{\Gamma \vdash \text{signal } x \text{ in } e : \dot{\tau} | D}$$

$$\frac{\Gamma(x) = \text{signal}\langle \dot{\tau}_x \rangle \quad \Gamma \vdash (e) : \dot{\tau}_x | D}{\Gamma \vdash \text{emit } x(e) : \text{unit} | D}$$

$$\frac{\Gamma(x) = \text{signal}\langle \dot{\tau} \rangle}{\Gamma \vdash ?x : \tau | \mathbf{0}}$$

syntactic sugar: **emit** $s \equiv \text{emit } s(\text{true})$ e.g. (**emit** $A; ?A$)

Valued signals

expression	$e ::= \dots$		signal x in e		emit $x(e)$		$?S$
evaluation context	$F ::= \dots$		emit $s(\square)$				
type	$\tau ::= \dots$		signal $\langle \dot{\tau} \rangle$				
basic type	$\dot{\tau} ::= \dots$						

$e/\mu \xrightarrow{S} e'/\mu'$: in the environment of signals S , the configuration e/μ reduces to e'/μ' .

		(not executable: need to guess v_x)
	$x \notin \text{dom}(S)$	$e/\mu \xrightarrow{S[x \mapsto v_x]} e'/\mu'$
signal x in $v/\mu \xrightarrow{S} v/\mu$	<hr/>	
	signal x in $e/\mu \xrightarrow{S}$	signal x in e'/μ'
	<hr/>	<hr/>
	$S(x) = v$	$S(x) = v$
emit $x(v)/\mu \xrightarrow{S} ()/\mu$		$?x/\mu \xrightarrow{S} v/\mu$

No signal status: values are encoded as bitvectors, each bit is absent (0) if not emitted

trap/exit, suspend and loop

expression $e ::= \dots \mid \text{trap } x \text{ in } e \mid \text{exit } x \mid \text{suspend } e \text{ when } x \mid \text{loop } e \text{ end}$
type $\tau ::= \dots \mid \text{trap}$
basic type $\hat{\tau} ::= \dots$

- ▶ Add typing rules:

$$\frac{\Gamma[x : \text{trap}] \vdash e : \text{unit} \mid D}{\Gamma \vdash \text{trap } x \text{ in } e : \text{unit} \mid D}$$

$$\frac{\Gamma(x) = \text{trap}}{\Gamma \vdash \text{exit } x : \text{unit} \mid \mathbf{0}}$$

$$\frac{\Gamma \vdash e : \hat{\tau} \mid D \quad \Gamma(x) = \text{signal} \langle \text{bool} \rangle}{\Gamma \vdash \text{suspend } e \text{ when } x : \hat{\tau} \mid D}$$

$$\frac{\Gamma \vdash e : \tau \mid \mathbf{1}}{\Gamma \vdash \text{loop } e \text{ end} : \text{unit} \mid \mathbf{1}}$$

- ▶ and reduction rules, e.g.:

$$\text{loop } e \text{ end} / \mu \xrightarrow{\text{S}} e; \text{loop } e \text{ end}$$

Defining ESTEREL derivated constructs and LUSTRE operators

Define ESTEREL derivated constructs in ECLAT

```
1  let halt() = loop pause end ;;
2  let sustain s = loop emit s; pause end ;;
3  let await s = trap T in loop pause; if ?s then exit T else () end ;;
4
5  let abort (f, s) =
6    trap T in [(suspend f() when s; exit T) || (await s; exit T)] ;;
7
8  let loop_each (f, s) = loop abort ((fun () -> (f(); halt())), s) end ;;
9
10 let abro (a, b, r, o) =
11   loop_each((fun () -> [await a || await b]; emit o), r) ;;
12
13 let abcro(a, b, c, r, o) =
14   signal t in [ abro(a, b, r, t) || abro(t, c, r, o) ] ;;
```

Define LUSTRE operators in ECLAT

```
1  let absent () = signal s in ?s ;;
2  let mux(a, b, c) = if a then b else c ;;
3
4  let fby (x, y) =
5    let (o, _) = reg (fun (_, prey) -> (prey, y)) init (absent(), x)
6    in o ;;
7
8  let when(f, clk) = if clk then f() else absent() ;;
9  let merge(clk, a, b) = if clk then a else b ;;
10
11 let fixpoint (f) =
12   signal s in emit s(f(?s)); ?s ;;
13
14  (for example *)
15 let sum i = fixpoint (fun s -> fby(0,s) + i) ;;
```

More syntactic sugar...

```
node  $f$   $p_i$  returns  $p_o$  =  
   $p_1 = e_1$ ;  
   $p_2 = e_2$ ;  
  ...  
   $p_n = e_n$ ;;
```

is encoded internally as:

```
let  $f$   $p_i$  =  
  let ( $p_1, p_2, \dots p_n$ ) = fixpoint (fun ( $p_1, p_2, \dots p_n$ )  $\rightarrow$  ( $e_1, e_2, \dots e_n$ ))  
  in  $p_o$  ;;
```

Also:

- ▶ infix notation for **fb**y
- ▶ e **when** e' stands for `when((fun () \rightarrow e), e')`

A toy example

Let's implement an accelerator, `collatz`, computing the *Collatz (Syracuse)* sequence ending with value 1:

$$\text{collatz}(i) = \begin{cases} 1 & \text{if } i = 1 \\ \text{collatz}(i/2) & \text{if } i \text{ is pair} \\ \text{collatz}(3 * i + 1) & \text{if } i \text{ is even} \end{cases}$$

In ECLAT, it is programmed as follows:

```
1 let rec collatz(i) =  
2   if i = 1 then i else  
3   if i mod 2 = 0 then collatz (i / 2)  
4   else collatz (3 * i + 1) ;;  
5
```

ECLAT

with a pause at each (direct or tail-recursive) call to function `collatz` (because these is defined with keyword `rec`).

6
7
8
9
10
11
12
13
14
15
16
17
18

```
[ let _ = collatz(a) in ...  
|| let _ = collatz(b) in ...
```

]

6

7

8

9

10

11

12

13

14

15

16

17

18

```
trap T in  
[ let _ = collatz(a) in exit T  
|| let _ = collatz(b) in exit T  
]
```

]

6
7
8
9
10
11
12
13
14
15
16
17
18

```
signal cy in ESTEREL  
  trap T in  
    [ let _ = collatz(a) in exit T  
    || let _ = collatz(b) in exit T  
    || loop emit cy(0 fby ?cy + 1) ; pause end ] )
```

6
7
8
9
10
11
12
13
14
15
16
17
18

```
(node main (a, b, button) returns (dur, occ, rdy) = LUSTRE
  (dur, rdy) =
    (exec signal cy in ESTEREL
      (trap T in
        [ let _ = collatz(a) in exit T
          || let _ = collatz(b) in exit T
          || loop emit cy(0 fby ?cy + 1) ; pause end ])
        ?cy
      default (0 fby dur)) ;;
```

6
7
8
9
10
11
12
13
14
15
16
17
18

```
(node main (a, b, button) returns (dur, occ, rdy) = LUSTRE
  occ = (0 fby occ + 1) when rdy ;

  (dur, rdy) =
    (exec signal cy in ESTEREL
      (trap T in
        [ let _ = collatz(a) in exit T
        || let _ = collatz(b) in exit T
        || loop emit cy(0 fby ?cy + 1) ; pause end ])
      ?cy
    default (0 fby dur)) ;;
```

6

7

8

9

10

11

12

13

14

15

16

17

18

```

(node main (a, b, button) returns (dur, occ, rdy) =
  occ    = (0 fby occ + 1) when rdy ;

  (dur, rdy) = if (a < 1) or (b < 1) then (0, false) else
    (exec signal cy in
      (trap T in
        [ let _ = collatz(a) in exit T
        || let _ = collatz(b) in exit T
        || loop emit cy(0 fby ?cy + 1) ; pause end ])
      ?cy
    default (0 fby dur)) ;

```

LUSTRE

ESTEREL

```

6  signal stop ;;
7
8  (node main (a, b, button) returns (dur, occ, rdy) =                                LUSTRE
9      occ      = (0 fby occ + 1) when rdy ;
10     ()       = if button then emit stop else () ;
11     (dur, rdy) = if (a < 1) or (b < 1) then (0, false) else
12                 (exec signal cy in                                             ESTEREL
13                     suspend (trap T in
14                         [ let _ = collatz(a) in exit T
15                           || let _ = collatz(b) in exit T
16                             || loop emit cy(0 fby ?cy + 1) ; pause end ])
17                     when stop; ?cy
18                 default (0 fby dur)) ; ;

```

Conclusion

- ▶ ECLAT design choices inspired by ESTEREL and LUSTRE (e.g., the non-blocking ECLAT construct **exec** for periodic execution of multicycle computations)
- ▶ the hope of being as expressive as ESTEREL and LUSTRE for programming reactive behaviors, while meeting the needs for high-level design of compute-intensive, timing-predictable and composable hardware accelerators.
- ▶ Experimental answer: a conservative extension of ECLAT with ESTEREL and LUSTRE constructs (taking inspiration from a lot of existing work in the community)
- ▶ But questions remain:
 - **checking causality** – at source level ? at circuit level ?
ECLAT is compiled to hierarchical mealy machines (keeping the control flow of the original program) rather than flat, easier to analyze netlists
 - recover a simple (ML-like) **clock calculus** while keeping dynamicity of signal broadcasting
 - **software simulation** and **executable semantics**
 - keep it simple, with respect to applicability and potential users